This file contains the computations relevant to the paper "On the classification of non-self-dual modular categories" by Hong and Rowell. This is the case where b=0, y=-f and a=g.

```
> with(LinearAlgebra): with(Groebner):
```

Using the Galois argument assuming that (01)(34), the orthogonality relations imply z=(d-1) and h1=-1/2(d+1).

```
> S1:=subs({x=1,y=-f,a=g,b=0},Matrix(5,5,[[1,d,f,g,g],[d,x,y,a,a],[f,y,z,b,b],[g,a,b,h1+I*h2,h1-I*h2],[g,a,b,h1-I*h2,h1+I*h2]]));
```

$$S1 := \begin{bmatrix} 1 & d & f & g & g \\ d & 1 & -f & g & g \\ f & -f & z & 0 & 0 \\ g & g & 0 & h1 + h2\, I & h1 - h2\, I \\ g & g & 0 & h1 - h2\, I & h1 + h2\, I \end{bmatrix}$$

```
> C:=Matrix(5,5,[[1,0,0,0,0],[0,1,0,0,0],[0,0,1,0,0],[0,0,0,0,1],[0,0,0,1,0]]);
```

$$C := \begin{bmatrix} 1 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 \\ 0 & 0 & 0 & 1 & 0 \end{bmatrix}$$

These relations describe the condition that S^2 is proportiional to the "charge conjugation matrix."

```
> Srels:=factor(convert(evalm(S1^2-K^2*C),set));
```

$$Srels := \{0, f\,(-z-1+d), (1+2\,h1+d)\,g, -f\,(-z-1+d), 2\,d-f^2+2\,g^2, 2\,f^2+z^2-K^2, 2\,g^2+2\,h1^2-2\,h2^2, 2\,g^2+2\,h1^2+2\,h2^2-K^2, 1+d^2+f^2+2\,g^2-K^2\}$$

```
> factor(Basis(`union`(Srels,{-z-1+d,1+2*h1+d}),lexdeg([K,d,f,g,z],[h2,h1])));
```

$$[2+z+2\,h1, 1+2\,h1+d, g^2+h1^2-h2^2, 2+4\,h1-2\,h2^2+2\,h1^2+f^2, (K-2\,h2)\,(K+2\,h2)]$$

```
> s1rules:=solve({2+z+2*h1, 1+2*h1+d},{z,h1});
```

$$s1rules := \{h1 = -\frac{d}{2}-\frac{1}{2}, z = d-1\}$$

```
> S:=subs(s1rules,S1);
```

$$S := \begin{bmatrix} 1 & d & f & g & g \\ d & 1 & -f & g & g \\ f & -f & d-1 & 0 & 0 \\ g & g & 0 & -\frac{d}{2}-\frac{1}{2}+h2\, I & -\frac{d}{2}-\frac{1}{2}-h2\, I \\ g & g & 0 & -\frac{d}{2}-\frac{1}{2}-h2\, I & -\frac{d}{2}-\frac{1}{2}+h2\, I \end{bmatrix}$$

These relations come down to a single relation among the d,f and g (the other allow elimination of K, h1 and h2 and z).

```
> orthrel:={-2*d+f^2-2*g^2};
```

$$orthrel := \{-2\, d + f^2 - 2\, g^2\}$$

Observe that if d is an integer, then d=1 since (01) interchanges d and 1/d. Thus f=-f/d=-f since f and -f/d is interchanged by (01), which implies f=0. So d is not an integer (for this case).

We use the various symmetries of the N_{i,j}^k to write down the fusion matrices in terms of just 14 variables. Note that M4 is just the transpose of M3.

```
> M1:=Matrix([[0,1,0,0,0],[1,n1,n2,n3,n3],[0,n2,n4,n5,n5],[0,n3,n5,n
  6,n7],[0,n3,n5,n7,n6]]);
```

$$M1 := \begin{bmatrix} 0 & 1 & 0 & 0 & 0 \\ 1 & n1 & n2 & n3 & n3 \\ 0 & n2 & n4 & n5 & n5 \\ 0 & n3 & n5 & n6 & n7 \\ 0 & n3 & n5 & n7 & n6 \end{bmatrix}$$

```
> M2:=Matrix([[0,0,1,0,0],[0,n2,n4,n5,n5],[1,n4,n8,n9,n9],[0,n5,n9,n
  10,n11],[0,n5,n9,n11,n10]]);
```

$$M2 := \begin{bmatrix} 0 & 0 & 1 & 0 & 0 \\ 0 & n2 & n4 & n5 & n5 \\ 1 & n4 & n8 & n9 & n9 \\ 0 & n5 & n9 & n10 & n11 \\ 0 & n5 & n9 & n11 & n10 \end{bmatrix}$$

```
> M3:=Matrix([[0,0,0,0,1],[0,n3,n5,n7,n6],[0,n5,n9,n11,n10],[1,n6,n1
  0,n12,n13],[0,n7,n11,n14,n12]]);
```

$$M3 := \begin{bmatrix} 0 & 0 & 0 & 0 & 1 \\ 0 & n3 & n5 & n7 & n6 \\ 0 & n5 & n9 & n11 & n10 \\ 1 & n6 & n10 & n12 & n13 \\ 0 & n7 & n11 & n14 & n12 \end{bmatrix}$$

The matrices must commute, giving a set of diophantine equations.

```
> comrels:=`minus`(`union`(convert(evalm(M1&*M2-M2&*M1),set),convert
  (evalm(M1&*M3-M3&*M1),set),convert(evalm(M2&*M3-M3&*M2),set),conve
  rt(evalm(M3&*Transpose(M3)-Transpose(M3)&*M3),set)),{0});
```

$comrels := \{n12 - n13, n13 - n12, n3\, n6 + n5\, n10 + n7\, n13 - n3\, n7 - n5\, n11 - n7\, n14,$

$n3\, n7 + n5\, n11 + n7\, n14 - n3\, n6 - n5\, n10 - n7\, n13,$

$n6\, n5 + n9\, n10 + n11\, n13 - n7\, n5 - n9\, n11 - n11\, n14,$

$n7\, n5 + n9\, n11 + n11\, n14 - n6\, n5 - n9\, n10 - n11\, n13, 1 + n6^2 + n10^2 + n13^2 - n7^2 - n11^2 - n14^2,$

$1 + n1\, n4 + n2\, n8 + 2\, n3\, n9 - n2^2 - n4^2 - 2\, n5^2, n2^2 + n4^2 + 2\, n5^2 - 1 - n1\, n4 - n2\, n8 - 2\, n3\, n9,$

$n3^2 + n5^2 + 2\, n7\, n6 - n1\, n7 - n2\, n11 - n3\, n12 - n3\, n14,$

$n5^2 + n9^2 + 2\, n11\, n10 - n4\, n7 - n11\, n8 - n9\, n12 - n9\, n14,$

$n7^2 + n11^2 + n14^2 - 1 - n6^2 - n10^2 - n13^2,$

$n1\, n7 + n2\, n11 + n3\, n12 + n3\, n14 - n3^2 - n5^2 - 2\, n7\, n6,$

$n4\, n7 + n11\, n8 + n9\, n12 + n9\, n14 - n5^2 - n9^2 - 2\, n11\, n10,$

$n1\, n5 + n2\, n9 + n3\, n10 + n3\, n11 - n3\, n2 - n5\, n4 - n6\, n5 - n7\, n5,$

$n2\, n5 + n4\, n9 + n5\, n10 + n5\, n11 - n3\, n4 - n5\, n8 - n6\, n9 - n7\, n9,$

$n2\, n6 + n4\, n10 + n5\, n13 + n5\, n12 - n3\, n5 - n5\, n9 - n6\, n10 - n7\, n11,$

$n2\, n7 + n4\, n11 + n5\, n12 + n5\, n14 - n3\, n5 - n5\, n9 - n6\, n11 - n7\, n10,$

$n3\, n2 + n5\, n4 + n6\, n5 + n7\, n5 - n1\, n5 - n2\, n9 - n3\, n10 - n3\, n11,$

$n3\, n4 + n5\, n8 + n6\, n9 + n7\, n9 - n2\, n5 - n4\, n9 - n5\, n10 - n5\, n11,$

$n3\, n5 + n5\, n9 + n6\, n10 + n7\, n11 - n2\, n6 - n4\, n10 - n5\, n13 - n5\, n12,$

$n3\, n5 + n5\, n9 + n6\, n11 + n7\, n10 - n2\, n7 - n4\, n11 - n5\, n12 - n5\, n14,$

$n3\, n6 + n5\, n10 + n6\, n13 + n7\, n12 - n3\, n7 - n5\, n11 - n6\, n12 - n7\, n14,$

$n3\, n7 + n5\, n11 + n6\, n12 + n7\, n14 - n3\, n6 - n5\, n10 - n6\, n13 - n7\, n12,$

$n6\, n5 + n9\, n10 + n10\, n13 + n11\, n12 - n7\, n5 - n9\, n11 - n10\, n12 - n11\, n14,$

$n7\, n5 + n9\, n11 + n10\, n12 + n11\, n14 - n6\, n5 - n9\, n10 - n10\, n13 - n11\, n12,$

$1 + n1\, n6 + n2\, n10 + n3\, n13 + n3\, n12 - n3^2 - n5^2 - n6^2 - n7^2,$

$1 + n4\, n6 + n8\, n10 + n9\, n13 + n9\, n12 - n5^2 - n9^2 - n10^2 - n11^2,$

$n3^2 + n5^2 + n6^2 + n7^2 - 1 - n1\, n6 - n2\, n10 - n3\, n13 - n3\, n12,$

$$n5^2+n9^2+n10^2+n11^2-1-n4\,n6-n8\,n10-n9\,n13-n9\,n12\}$$

```
> indets(comrels);
```

$$\{n1, n10, n11, n12, n13, n14, n2, n3, n4, n5, n6, n7, n8, n9\}$$

The commutation relations alone leave 6 degrees of freedom.

```
> HilbertDimension(comrels,tdeg(n1, n10, n11, n12, n13, n14, n2, n3,
  n4, n5, n6, n7, n8, n9));
```

$$6$$

The Characteristic polynomials of M1,M2 and M3 are computed in two ways. First from the knowledge of the eigenvalues.

```
> ch1:=collect(expand(product('(X-S[2,i]/S[1,i])',i=1..5)),X);
  ch2:=collect(expand(product('(X-S[3,i]/S[1,i])',i=1..5)),X);
  ch3:=collect(expand(product('(X-S[4,i]/S[1,i])',i=1..5)),X);
```

$$ch1 := 1+X^5+\left(-\frac{1}{d}-1-d\right)X^4+\left(\frac{1}{d}+d\right)X^3+\left(\frac{1}{d}+d\right)X^2+\left(-\frac{1}{d}-1-d\right)X$$

$$ch2 := X^5+\left(\frac{1}{f}+\frac{f}{d}-\frac{d}{f}-f\right)X^4+\left(-2+d-\frac{f^2}{d}+\frac{1}{d}\right)X^3+\left(f-\frac{f}{d}\right)X^2$$

$$ch3 := X^5+\left(-g+\frac{1}{g}+\frac{d}{g}-\frac{g}{d}\right)X^4+\left(\frac{g^2}{d}-\frac{1}{d}+\frac{d}{2\,g^2}-2+\frac{h2^2}{g^2}-d+\frac{1}{4\,g^2}+\frac{d^2}{4\,g^2}\right)X^3$$
$$+\left(-\frac{h2^2}{g}-\frac{3}{4\,g}+g-\frac{d^2}{4\,g}-\frac{h2^2}{d\,g}-\frac{1}{4\,d\,g}+\frac{g}{d}-\frac{3\,d}{4\,g}\right)X^2+\left(\frac{1}{2}+\frac{1}{4\,d}+\frac{d}{4}+\frac{h2^2}{d}\right)X$$

```
>
```

set m=(d-1)/f which in an integer.

```
> simplify({subs({m=(d-1)/f},coeff(ch2,X,4)+coeff(ch2,X,2)+m),subs({
  m=(d-1)/f},subs({X=m},ch2))});
```

$$\{0\}$$

The form of these polynomials imply further relations.

```
> simplify({coeff(ch1,X,0)-1,coeff(ch2,X,0),coeff(ch2,X,1),coeff(ch3
  ,X,0),coeff(ch1,X,4)-coeff(ch1,X,1)});
```

$$\{0\}$$

```
> factor(ch1); factor(ch2); factor(ch3);
```

$$\frac{(X+1)\,(X-1)^2\,(d\,X-1)\,(X-d)}{d}$$

$$\frac{X^2\,(X-f)\,(f\,X+1-d)\,(f+d\,X)}{d\,f}$$

$$\frac{X\,(X-g)\,(d\,X-g)\,(4\,X^2\,g^2+4\,g\,X\,d+4\,g\,X+2\,d+1+d^2+4\,h2^2)}{4\,d\,g^2}$$

```
> p1:=CharacteristicPolynomial(M1,X):
  p2:=CharacteristicPolynomial(M2,X):
  p3:=CharacteristicPolynomial(M3,X):
> factor(p1); factor(p2);
```

$$(X-n6+n7)\,(-X^3\,n4-X^3\,n6-2\,X^2\,n5^2+2\,n5^2-n4\,n7-n4\,n6-X^2+X^4+X^2\,n1\,n6+X^2\,n1\,n4$$

$$+X^2\, n1\, n7+X^2\, n4\, n7+2\, X\, n1\, n5^2+2\, X\, n3^2\, n4+X\, n6\, n2^2+X\, n2^2\, n7-X^3\, n1-2\, X^2\, n3^2 -X^2\, n2^2-X^3\, n7+X\, n4+X\, n6+X\, n7-4\, X\, n2\, n3\, n5-X\, n6\, n1\, n4-X\, n1\, n4\, n7+X^2\, n6\, n4)$$

$$(X+n11-n10)\,(-X^3\, n11-2\, X^2\, n5^2+2\, n5^2-n2\, n11-n2\, n10-X^2+X^4+2\, X\, n5^2\, n8 +2\, X\, n2\, n9^2+X\, n10\, n4^2+X\, n11\, n4^2+X\, n10+X^2\, n11\, n8+X^2\, n2\, n8+X^2\, n2\, n10+X^2\, n2\, n11 -2\, X^2\, n9^2-X^3\, n10-X^3\, n8-X^3\, n2-X^2\, n4^2+X\, n2+X^2\, n10\, n8-4\, X\, n4\, n5\, n9-X\, n2\, n8\, n11 -X\, n10\, n2\, n8+X\, n11)$$

Observe that the only linear terms of p1 are (X-1) and (X+1), while the linear terms of p2 are X and (X-m)
we may use this later, but for now we avoid using m.

```
> solve(X+n11-n10=(X-m)); solve(X+n11-n10=X);
```

$$\{X=X,\, m=-n11+n10,\, n10=n10,\, n11=n11\}$$

$$\{X=X,\, n10=n11,\, n11=n11\}$$

```
> mlinrels:={(n10-n11)*(n10-n11-m)};
```

$$mlinrels:=\{(-n11+n10)\,(-m-n11+n10)\}$$

```
> linrels:={(n6-n7)^2-1};
```

$$linrels:=\{(-n7+n6)^2-1\}$$

The relations implied by the coefficients of the Characteristic polynomials.

```
> chrels:=factor({coeff(p1,X,0)-1,coeff(p2,X,0),coeff(p2,X,1),coeff(
  p3,X,0),coeff(p1,X,4)-coeff(p1,X,1)});
```

$$chrels:=\{(-n11+n10)\,(-2\, n5^2+n2\, n11+n2\, n10),\, n4\, n6^2-n4\, n7^2-2\, n5^2\, n6+2\, n5^2\, n7-1, n7^2\, n9-2\, n11\, n5\, n7+n14\, n5^2-n14\, n3\, n9+n11^2\, n3,\, 4\, n10\, n4\, n5\, n9-2\, n2\, n10+2\, n11\, n5^2\, n8 -4\, n5\, n4\, n9\, n11+2\, n5^2-2\, n5^2\, n8\, n10-2\, n9^2\, n2\, n10+n11^2\, n4^2-n10^2\, n4^2-n11^2\, n2\, n8 +2\, n9^2\, n11\, n2+n11^2+n10^2\, n2\, n8-n10^2,\, -n4-n1-2\, n6+n6^2-n7^2+n6^2\, n2^2 -4\, n2\, n3\, n5\, n6-n1\, n4\, n6^2-2\, n1\, n5^2\, n7+4\, n2\, n3\, n5\, n7+n1\, n4\, n7^2+2\, n1\, n5^2\, n6 -2\, n3^2\, n7\, n4+2\, n3^2\, n4\, n6+2\, n4\, n6-n7^2\, n2^2-2\, n5^2\}$$

```
> nops(chrels);
```

$$5$$

these involve the integer m=z/f=(d-1)/f.

```
> intchar:=Vector([1,-1,m,0,0]):
> intcharrels:=`union`(convert(evalm(M1&*intchar+intchar),set),conve
  rt(evalm(M2&*intchar-m*intchar),set),convert(evalm(M3&*intchar),se
  t),convert(evalm(Transpose(M3)&*intchar),set));
```

$$intcharrels:=\{0,\, -n1+n2\, m,\, -n3+n5\, m,\, -n5+n9\, m,\, -n7+n11\, m,\, 1-n6+n10\, m, -n2+n4\, m+m,\, 1-n4+n8\, m-m^2\}$$

```
> mrels:=`union`({coeff(p2,X,4)+coeff(p2,X,2)+m,subs({X=m},p2)},mlin
  rels,intcharrels);
```

$$mrels:=\{0,\, (-n11+n10)\,(-m-n11+n10),\, -n1+n2\, m,\, -n3+n5\, m,\, -n5+n9\, m,\, -n7+n11\, m, 1-n6+n10\, m,\, -n2+n4\, m+m,\, 1-n4+n8\, m-m^2,\, m^5-(2\, n10+n8+n2)\, m^4 -(2\, n5^2+2\, n9^2+n11^2-n10^2-2\, n8\, n10-2\, n2\, n10+1+n4^2-n2\, n8)\, m^3-(-n11^2\, n8-n2$$

$$-2\,n10-2\,n5^2\,n8+2\,n11\,n5^2+4\,n4\,n5\,n9-2\,n2\,n9^2-2\,n10\,n4^2+2\,n9^2\,n11-2\,n10\,n5^2-2\,n10\,n9^2-n11^2\,n2+n8\,n10^2+n2\,n10^2+2\,n10\,n2\,n8)\,m^2-(-n11^2-2\,n5^2+2\,n2\,n10+n10^2-2\,n11\,n5^2\,n8+4\,n5\,n4\,n9\,n11-4\,n10\,n4\,n5\,n9-n11^2\,n4^2+n10^2\,n4^2-2\,n9^2\,n11\,n2+2\,n5^2\,n8\,n10+2\,n9^2\,n2\,n10+n11^2\,n2\,n8-n10^2\,n2\,n8)\,m+2\,n11\,n5^2-2\,n10\,n5^2-n11^2\,n2+n2\,n10^2,\ -n8+n11^2\,n8+2\,n5^2\,n8-2\,n11\,n5^2-4\,n4\,n5\,n9+2\,n2\,n9^2+2\,n10\,n4^2-2\,n9^2\,n11+2\,n10\,n5^2+2\,n10\,n9^2+n11^2\,n2-n8\,n10^2-n2\,n10^2-2\,n10\,n2\,n8+m\}$$

We take all of the relations together and process them. .

```
> allNrels:=subs({n8=u,n11=v,n13=t,n14=w},`union`(linrels,comrels,chrels,mrels)):
>
> factor(Basis(allNrels,lexdeg([w, v, t, u,n1,n6, n9,n10,n12, n2, n3, n4, n5,n7],[m])));
```

$$[(2+m^2)\,(-1-2\,n7+n4),\ n3-n5\,m,\ n2-n4\,m-m,\ 1-n4+2\,n10\,m,\ -n5+n9\,m,\ -n4-1+2\,n6,\ -2+n1+2\,n4-4\,n7-2\,m^2-2\,n7\,m^2,\ 1-n4+u\,m-m^2,\ t-n12,\ -n7+v\,m,\ (2+m^2)\,(m^2\,n5+m\,n12+w\,m+2\,n7\,m\,n12-n5-n7\,n5),\ 2\,n9-2\,n12-2\,w-2\,n12\,n7\,m^2+2\,n7\,n9-n5\,m-4\,n7\,n12-m^3\,n5+n7\,n5\,m-w\,m^2-m^2\,n12,\ -n7\,(m+2\,n10-u),\ -n7\,(-2\,n7\,m+4\,n10+n4\,m-m-4\,v),\ -(2+m^2)\,(n5\,m+n7\,n12-n7\,w),\ -1+n4+2\,n5^2+2\,n7^2-3\,n7-3\,n4\,n7,\ -n5-2\,n7\,m\,n12+n5\,n4-2\,n7\,n5-2\,m^2\,n5+2\,n7\,w\,m,\ -4\,n10-8\,v+2\,u+3\,m-8\,n7\,n10+3\,n4\,m+n7\,m+n7\,m\,n4-2\,n7^2\,m+8\,n5\,n12,\ 2\,n5\,n10+2\,n9-2\,n12-2\,w-2\,n12\,n7\,m^2-3\,n5\,m-6\,n7\,n12-m^3\,n5+n7\,n5\,m-w\,m^2-m^2\,n12+2\,n7\,w,\ -n4\,m-m-4\,n10-8\,v+2\,u+4\,n5\,n9-8\,n7\,n10+n7\,m\,n4+n7\,m-2\,n7^2\,m,\ 2\,n9-2\,n12-2\,w-2\,n12\,n7\,m^2-4\,n5\,m-6\,n7\,n12-m^3\,n5+n7\,n5\,m-w\,m^2-m^2\,n12+2\,n7\,w+n5\,u,\ 2\,n9-2\,n12-2\,w-2\,n12\,n7\,m^2-n5\,m-4\,n7\,n12-m^3\,n5+n7\,n5\,m-w\,m^2-m^2\,n12+2\,n5\,v,\ -n4\,m-9\,m-4\,n10-8\,v+2\,u-8\,n7\,n10-3\,n7\,m\,n4-3\,n7\,m+8\,n5\,w+6\,n7^2\,m,\ (n4+3-2\,n7)\,(-1-2\,n7+n4),\ (-1-2\,n7+n4)\,(-2\,m+n7\,m+2\,n10),\ n9+n4\,n9-2\,n12-2\,w-2\,n12\,n7\,m^2-3\,n5\,m-6\,n7\,n12-m^3\,n5+n7\,n5\,m-w\,m^2-m^2\,n12+2\,n7\,w,\ -3\,n4\,m+3\,m-u-4\,n7\,n10+n7\,m\,n4+3\,n7\,m+u\,n4-2\,n7^2\,m,\ -v-2\,n7\,n10+n4\,v,\ -(-1-2\,n7+n4)\,(n12-w),\ 1-n4-4\,n10^2+2\,u\,n10,\ -2+2\,n4-3\,n7+2\,n7^2-4\,n10^2-n4\,n7+4\,v\,n10,\ -1+n9^2-n4-2\,n9\,n12,\ n5-2\,n9\,n10+2\,n10\,n12+3\,m\,n12-2\,w\,n10+u\,n9-2\,n12\,u+m^2\,n5+3\,n7\,m\,n12+4\,v\,n12-m\,n12\,n4-n7\,w\,m,\ -4\,n9\,n10+2\,n10\,n12-3\,m^2\,n5-5\,n7\,m\,n12-m\,n12+4\,n9\,v+2\,w\,n10+3\,n7\,w\,m-2\,n5-4\,v\,n12+m\,n12\,n4,\ -3-n4-n7+2\,n7^2-n4\,n7+2\,n9\,w-2\,n9\,n12,\ 2-2\,n4-4\,n10^2-m^2+u^2,\ -2+2\,u\,v+2\,n4-5\,n7+2\,n7^2-4\,n10^2-n4\,n7,\ 2\,n5+m^2\,n5+3\,n7\,m\,n12+2\,m\,n12+4\,v\,n12-4\,w\,n10-w\,m-m\,n12\,n4-n7\,w\,m-n12\,u+u\,w,$$

$-1+n4+2\,v^2+2\,n7^2-2\,n10^2-n7-n4\,n7,$

$-2\,n5-2\,n10\,n12-m^2\,n5+n7\,m\,n12+4\,v\,w-2\,w\,n10+m\,n12-m\,n12\,n4+n7\,w\,m,$

$-3-n4-n7+2\,n7^2-n4\,n7+2\,w^2-2\,n12^2, 3\,n5+2\,n7\,n5-n5\,n7^2+2\,n12\,n7^2\,m+m^2\,n5$
$+3\,n7\,m\,n12-m\,n12+n7\,m^2\,n5-3\,w\,m-2\,m\,n12\,n4-3\,n7\,w\,m, -n12-2\,n12\,n7\,m^2-2\,n5\,m$
$-6\,n7\,n12-m^3\,n5-w\,m^2-m^2\,n12+n12\,n4+3\,n7\,w+w\,n7^2-3\,n12\,n7^2+n7\,n12\,n4,$

$2-2\,n4+3\,n7+4\,n12\,n7\,w+3\,n7^2-3\,n4\,n7+2\,n7^3-n4\,n7^2-4\,n12^2\,n7+4\,m^2+4\,n7\,m^2, 1$
$-n4-13\,n7-4\,m^2\,n12\,w-18\,n7^2+3\,n4\,n7+8\,n9\,n12+8\,n10^2-8\,n12^2-4\,m^2\,n12^2+4\,m^4\,n7$
$-4\,n7^3-8\,n7^2\,m^2+4\,n4\,n7^2+8\,n7\,n10^2+4\,m^2-8\,n12^2\,n7\,m^2+4\,m^4-4\,n7\,m^2-16\,n12^2\,n7$
$-8\,w\,n12, 3\,n5-2\,n7\,n5-n5\,n7^2+m^2\,n5+3\,n7\,m\,n12-4\,n10\,n12\,n7-m\,n12+4\,w\,n10\,n7$
$-n7\,m^2\,n5-3\,w\,m-2\,m\,n12\,n4-3\,n7\,w\,m+2\,n7^2\,w\,m, -2\,n7^3\,m+20\,n10-24\,v-2\,u+4\,m$
$-8\,n7\,n10+8\,n10\,n12^2-8\,n7\,m\,n12^2+4\,m\,n12^2\,n4+6\,n4\,m-5\,n7\,m+8\,w\,n10\,n12+n7\,m\,n4$
$+m\,n4\,n7^2-16\,n12^2\,v-4\,m\,n12^2-3\,n7^2\,m, -n5\,m-m^3\,n5-2\,n12-2\,w-m^2\,n12+2\,n9$
$-w\,m^2-4\,n7\,n12-2\,n12\,n7\,m^2-n12\,n7^2+w\,n7^2-2\,n10^2\,n12+2\,w\,n10^2]$

m>0 an integer implies:.

```
> factor(Basis(‘union‘(allNrels,{-1-2*n7+n4,m^2*n5+m*n12+w*m+2*n7*m*
  n12-n5-n7*n5,n5*m+n7*n12-n7*w}),plex(n1,n6, n9,n12, n2, n3, n4,w,
  v,n10, t, u, n5,n7,m)));
```

$[n5^2-2\,n7^2-2\,n7, -2\,n7-m^2+u\,m, (1+n7)\,(4\,n7\,t+3\,n5\,m-n5\,u),$

$-n7-n7^2+n5\,m\,t+m^2+n7\,m^2, n7\,(3\,m-u+2\,n5\,t-u\,n7+3\,n7\,m),$

$6\,n7+3\,m^2-u^2-u^2\,n7+2\,u\,t\,n5+6\,n7^2+3\,n7\,m^2, -n7+n10\,m, n7\,(m+2\,n10-u),$

$n5\,(m+2\,n10-u), 2\,n7+m^2+2\,u\,n10-u^2, 4\,n7+4\,n10^2+m^2-u^2,$

$-6\,n7\,m-3\,m+2\,n10+4\,v-u-4\,n5\,t+2\,u\,n7, -n5+m^2\,n5-n7\,n5+2\,n7\,m\,t+w\,m+t\,m,$

$-n5\,m-n7\,t+n7\,w, -2\,n7\,m-2\,m-n5\,t+n5\,w,$

$-3\,n5\,m-2\,n5\,u+2\,t\,m^2-n7\,n5\,m-u^2\,t+m^3\,n5+2\,n7\,m^2\,t+u^2\,w, -8\,m-8\,u+32\,n10$
$-19\,n7\,m+7\,u\,n7-22\,n5\,t-2\,m^3+5\,n7^2\,m-2\,t^2\,m-u\,n7^2-4\,u\,t^2+12\,n10\,t^2-2\,n7\,m^3$
$-8\,n7\,m\,t^2-8\,t^3\,n5+4\,u\,n7\,t^2+2\,t\,w\,u,$

$-n5-5\,t\,m+n7\,n5+2\,n10\,t-m^2\,n5-8\,n7\,m\,t-4\,t^2\,n5+2\,u\,t\,n7-u\,w+4\,w\,n10,$

$-2\,n7-2-t^2+w^2, -1-2\,n7+n4, n3-n5\,m, n2-2\,n7\,m-2\,m, n12-t,$

$-3\,n5\,m+2\,n9-2\,t-2\,w-4\,n7\,t+n5\,u, n6-n7-1, n1-2\,n7\,m^2-2\,m^2]$

assuming n7=0 gives:

```
> factor(Basis(‘union‘(allNrels,{-1-2*n7+n4,m^2*n5+m*n12+w*m+2*n7*m*
  n12-n5-n7*n5,n5*m+n7*n12-n7*w,n7}),plex(n1,n6, n9,n10,n12, n2, n3,
  n4,w, v, t, u, n5,n7,m)));
```

$[m^2, n7, n5\,m, n5^2, u\,m, n5\,u, u^2, v\,m, n5\,v, u\,v, v^2, w\,m+t\,m-n5, -n5\,t-2\,m+n5\,w,$

$-32\,v+4\,u+20\,m+21\,n5\,t-12\,t^2\,v+u\,t^2+8\,t^3\,n5+8\,t^2\,m+t\,w\,u,$

$4\,v\,t-u\,t-5\,n5-8\,t^2\,n5-8\,t\,m+8\,v\,w-u\,w,\ w^2-2-t^2,\ n4-1,\ n3,\ n2-2\,m,\ n12-t,$
$-4\,n5\,t-u+4\,v+2\,n10-3\,m,\ n9-w-t,\ n6-1,\ n1]$

Thus we may assume n7 not zero, since then we have m=0 and d=1. so we see that -u+2*n10+m=0,m+2*v-u=0

```
> factor(Basis(‘union‘(allNrels,{-1-2*n7+n4,m^2*n5+m*n12+w*m+2*n7*m*
  n12-n5-n7*n5,n5*m+n7*n12-n7*w,-u+2*n10+m,m+2*v-u}),plex(n1,n6,
  n9,n10,n12, n2, n3, n4,w, v, t, u, n5,n7,m)));
```

$[n5^2-2\,n7^2-2\,n7,\ -2\,n7-m^2+u\,m,\ (1+n7)\,(4\,n7\,t+3\,n5\,m-n5\,u),$
$3\,m-u+2\,n5\,t-u\,n7+3\,n7\,m,\ m+2\,v-u,\ -n5+m^2\,n5-n7\,n5+2\,n7\,m\,t+w\,m+t\,m,$
$-n5\,m-n7\,t+n7\,w,\ -m-u-n7\,m-u\,n7+2\,n5\,w,$
$-3\,n5+2\,t\,m-u\,t-n7\,n5+m^2\,n5+2\,n7\,m\,t+u\,w,\ -2\,n7-2-t^2+w^2,\ -1-2\,n7+n4,\ n3-n5\,m,$
$n2-2\,n7\,m-2\,m,\ n12-t,\ m+2\,n10-u,\ -3\,n5\,m+2\,n9-2\,t-2\,w-4\,n7\,t+n5\,u,\ n6-n7-1,$
$n1-2\,n7\,m^2-2\,m^2]$

n7+1 not 0. also observe that we have a diophantine equation in n5 and n7, which has n5=0 iff n7=0. So n5 is not zero.

```
> intrels:=factor(Basis(‘union‘(allNrels,{-1-2*n7+n4,m^2*n5+m*n12+w*
  m+2*n7*m*n12-n5-n7*n5,n5*m+n7*n12-n7*w,-u+2*n10+m,m+2*v-u,4*n7*t+3
  *n5*m-n5*u}),lexdeg([m,n1,n6, n9,n10,n12, n2, n3, n4,n5,n7],[w, v,
  t, u])));
```

$intrels := [-u\,t+v\,t-u\,w+3\,v\,w,\ -2-t^2-2\,u\,v+4\,v^2+w^2,$
$3\,u-9\,v+u\,t^2-4\,t^2\,v+3\,u^2\,v-15\,v^2\,u+18\,v^3+t\,w\,u,\ n7+2\,v^2-u\,v,\ 3\,n5+4\,v\,t-u\,w-u\,t,$
$-1-2\,u\,v+4\,v^2+n4,\ -u^2\,t+16\,v\,u\,t-24\,t\,v^2-u^2\,w+9\,n3,$
$-6\,u-6\,u^2\,v+4\,u\,t^2+4\,t\,w\,u+12\,v^2\,u-16\,t^2\,v+9\,n2,\ n12-t,\ -v+n10,\ n9-w-t,$
$-1-u\,v+2\,v^2+n6,\ -6\,u^2-6\,u^3\,v+8\,u^2\,t^2+8\,w\,u^2\,t+12\,u^2\,v^2-80\,u\,t^2\,v+96\,t^2\,v^2+27\,n1,$
$m+2\,v-u]$

```
> rules1:=‘union‘({n8=u,n11=v,n13=t,n14=w},solve({seq(intrels[i],i=4
  ..14)}));
```

$$rules1 := \{m=-2\,v+u,\ n1=\frac{2\,u^2}{9}+\frac{2\,u^3\,v}{9}-\frac{8\,u^2\,t^2}{27}-\frac{8\,w\,u^2\,t}{27}-\frac{4\,u^2\,v^2}{9}+\frac{80\,u\,t^2\,v}{27}-\frac{32\,t^2\,v^2}{9},$$
$$n10=v,\ n11=v,\ n12=t,\ n13=t,\ n14=w,\ n2=\frac{2\,u}{3}+\frac{2\,u^2\,v}{3}-\frac{4\,u\,t^2}{9}-\frac{4\,t\,w\,u}{9}-\frac{4\,v^2\,u}{3}+\frac{16\,t^2\,v}{9},$$
$$n3=\frac{u^2\,t}{9}-\frac{16\,v\,u\,t}{9}+\frac{8\,t\,v^2}{3}+\frac{u^2\,w}{9},\ n4=1+2\,u\,v-4\,v^2,\ n5=-\frac{4\,v\,t}{3}+\frac{u\,w}{3}+\frac{u\,t}{3},$$
$$n6=1+u\,v-2\,v^2,\ n7=-2\,v^2+u\,v,\ n8=u,\ n9=t+w,\ t=t,\ u=u,\ v=v,\ w=w\}$$

```
> indets(subs(rules1,‘union‘(allNrels,{-1-2*n7+n4,m^2*n5+m*n12+w*m+2
  *n7*m*n12-n5-n7*n5,n5*m+n7*n12-n7*w,-u+2*n10+m,m+2*v-u,4*n7*t+3*n5
  *m-n5*u})));
```

$$\{t,\ u,\ v,\ w\}$$

```
> agb:=factor(Basis(subs(rules1,‘union‘(allNrels,{-1-2*n7+n4,m^2*n5+
```

```
m*n12+w*m+2*n7*m*n12-n5-n7*n5,n5*m+n7*n12-n7*w,-u+2*n10+m,m+2*v-u,
4*n7*t+3*n5*m-n5*u})),plex( u, v, t,w )));
```

agb :=

$$[2\,t\,v^2-t^3-2\,w\,v^2-t^2\,w+t\,w^2+w^3-2\,t-2\,w,\,u\,t-v\,t+u\,w-3\,v\,w,\,2+t^2+2\,u\,v-4\,v^2-w^2]$$

```
> M1s:=subs(rules1,M1); M2s:=subs(rules1,M2); M3s:=subs(rules1,M3);
```

M1s :=

$$[0\,,1\,,0\,,0\,,0]$$

$$\left[1\,,\frac{2\,u^2}{9}+\frac{2\,u^3\,v}{9}-\frac{8\,u^2\,t^2}{27}-\frac{8\,w\,u^2\,t}{27}-\frac{4\,u^2\,v^2}{9}+\frac{80\,u\,t^2\,v}{27}-\frac{32\,t^2\,v^2}{9}\,,\right.$$
$$\frac{2\,u}{3}+\frac{2\,u^2\,v}{3}-\frac{4\,u\,t^2}{9}-\frac{4\,t\,w\,u}{9}-\frac{4\,v^2\,u}{3}+\frac{16\,t^2\,v}{9}\,,\frac{u^2\,t}{9}-\frac{16\,v\,u\,t}{9}+\frac{8\,t\,v^2}{3}+\frac{u^2\,w}{9}\,,$$
$$\left.\frac{u^2\,t}{9}-\frac{16\,v\,u\,t}{9}+\frac{8\,t\,v^2}{3}+\frac{u^2\,w}{9}\right]$$

$$\left[0\,,\frac{2\,u}{3}+\frac{2\,u^2\,v}{3}-\frac{4\,u\,t^2}{9}-\frac{4\,t\,w\,u}{9}-\frac{4\,v^2\,u}{3}+\frac{16\,t^2\,v}{9}\,,1+2\,u\,v-4\,v^2\,,-\frac{4\,v\,t}{3}+\frac{u\,w}{3}+\frac{u\,t}{3}\,,\right.$$
$$\left.-\frac{4\,v\,t}{3}+\frac{u\,w}{3}+\frac{u\,t}{3}\right]$$

$$\left[0\,,\frac{u^2\,t}{9}-\frac{16\,v\,u\,t}{9}+\frac{8\,t\,v^2}{3}+\frac{u^2\,w}{9}\,,-\frac{4\,v\,t}{3}+\frac{u\,w}{3}+\frac{u\,t}{3}\,,1+u\,v-2\,v^2\,,-2\,v^2+u\,v\right]$$

$$\left[0\,,\frac{u^2\,t}{9}-\frac{16\,v\,u\,t}{9}+\frac{8\,t\,v^2}{3}+\frac{u^2\,w}{9}\,,-\frac{4\,v\,t}{3}+\frac{u\,w}{3}+\frac{u\,t}{3}\,,-2\,v^2+u\,v\,,1+u\,v-2\,v^2\right]$$

M2s :=

$$[0\,,0\,,1\,,0\,,0]$$

$$\left[0\,,\frac{2\,u}{3}+\frac{2\,u^2\,v}{3}-\frac{4\,u\,t^2}{9}-\frac{4\,t\,w\,u}{9}-\frac{4\,v^2\,u}{3}+\frac{16\,t^2\,v}{9}\,,1+2\,u\,v-4\,v^2\,,-\frac{4\,v\,t}{3}+\frac{u\,w}{3}+\frac{u\,t}{3}\,,\right.$$
$$\left.-\frac{4\,v\,t}{3}+\frac{u\,w}{3}+\frac{u\,t}{3}\right]$$

$$[1\,,1+2\,u\,v-4\,v^2\,,u\,,t+w\,,t+w]$$

$$\left[0\,,-\frac{4\,v\,t}{3}+\frac{u\,w}{3}+\frac{u\,t}{3}\,,t+w\,,v\,,v\right]$$

$$\left[0\,,-\frac{4\,v\,t}{3}+\frac{u\,w}{3}+\frac{u\,t}{3}\,,t+w\,,v\,,v\right]$$

$$M3s := \begin{bmatrix} 0 & 0 & 0 & 0 & 1 \\ 0 & \frac{u^2 t}{9} - \frac{16 v u t}{9} + \frac{8 t v^2}{3} + \frac{u^2 w}{9} & -\frac{4 v t}{3} + \frac{u w}{3} + \frac{u t}{3} & -2 v^2 + u v & 1 + u v - 2 v^2 \\ 0 & -\frac{4 v t}{3} + \frac{u w}{3} + \frac{u t}{3} & t + w & v & v \\ 1 & 1 + u v - 2 v^2 & v & t & t \\ 0 & -2 v^2 + u v & v & w & t \end{bmatrix}$$

```
>
```

The unique positive character is psi0.

```
> psi0:=Vector([1,d,f,g,g]): psi2:=subs({m=(d-1)/f},intchar):
> psirels:=numer(‘minus‘(‘union‘(convert(evalm(M1s&*psi0-psi0[2]*psi
  0),set),convert(evalm(M2s&*psi0-psi0[3]*psi0),set),convert(evalm(M
  3s&*psi0-psi0[4]*psi0),set),convert(evalm(M1s&*psi2-psi2[2]*psi2),
  set),convert(evalm(M2s&*psi2-psi2[3]*psi2),set),convert(evalm(M3s&
  *psi2-psi2[4]*psi2),set)),{0}));
```

$psirels := \{v\,(-u\,f + 2\,v\,f + d - 1),\ -2\,d\,v^2 + d\,u\,v + v\,f + g\,w + t\,g - g^2,$
$1 + d + d\,u\,v - 2\,d\,v^2 + v\,f + 2\,t\,g - g^2,\ -4\,d\,v\,t + d\,u\,w + d\,u\,t + 3\,f\,t + 3\,f\,w + 6\,v\,g - 3\,f\,g,$
$-u\,t\,f + 4\,v\,t\,f - u\,w\,f + 3\,t\,d - 3\,t + 3\,w\,d - 3\,w,\ -2\,u\,v\,f^2 + 4\,v^2\,f^2 + u\,f\,d - u\,f - d^2 + 2\,d - 1,$
$1 + d + 2\,d\,u\,v - 4\,d\,v^2 + u\,f + 2\,t\,g + 2\,g\,w - f^2,$
$16\,v\,u\,t\,f - 24\,t\,v^2\,f - u^2\,t\,f - u^2\,w\,f - 12\,d\,v\,t + 12\,v\,t + 3\,d\,u\,w - 3\,u\,w + 3\,d\,u\,t - 3\,u\,t,$
$d\,u^2\,t - 16\,d\,v\,u\,t + 24\,d\,t\,v^2 + d\,u^2\,w - 12\,v\,t\,f + 3\,u\,w\,f + 3\,u\,t\,f + 9\,g + 18\,g\,u\,v - 36\,g\,v^2 - 9\,d\,g,$
$-6\,u\,f - 6\,f\,u^2\,v + 4\,f\,u\,t^2 + 4\,f\,t\,w\,u + 12\,f\,v^2\,u - 16\,f\,t^2\,v + 18\,d - 18 + 18\,d\,u\,v - 18\,u\,v - 36\,d\,v^2$
$+ 36\,v^2,\ 6\,d\,u + 6\,d\,u^2\,v - 4\,d\,u\,t^2 - 4\,d\,t\,w\,u - 12\,d\,v^2\,u + 16\,d\,t^2\,v + 9\,f + 18\,u\,v\,f - 36\,v^2\,f$
$- 24\,g\,v\,t + 6\,g\,u\,w + 6\,g\,u\,t - 9\,d\,f,\ 27 + 6\,d\,u^2 + 6\,d\,u^3\,v - 8\,d\,u^2\,t^2 - 8\,d\,w\,u^2\,t - 12\,d\,u^2\,v^2$
$+ 80\,d\,u\,t^2\,v - 96\,d\,t^2\,v^2 + 18\,u\,f + 18\,f\,u^2\,v - 12\,f\,u\,t^2 - 12\,f\,t\,w\,u - 36\,f\,v^2\,u + 48\,f\,t^2\,v + 6\,g\,u^2\,t$
$- 96\,g\,v\,u\,t + 144\,g\,t\,v^2 + 6\,g\,u^2\,w - 27\,d^2,\ -6\,u^2\,f - 6\,u^3\,v\,f + 8\,u^2\,t^2\,f + 8\,w\,u^2\,t\,f + 12\,u^2\,v^2\,f$
$- 80\,u\,t^2\,v\,f + 96\,t^2\,v^2\,f + 18\,d\,u - 18\,u + 18\,d\,u^2\,v - 18\,u^2\,v - 12\,d\,u\,t^2 + 12\,u\,t^2 - 12\,d\,t\,w\,u$
$+ 12\,t\,w\,u - 36\,d\,v^2\,u + 36\,v^2\,u + 48\,d\,t^2\,v - 48\,t^2\,v\}$

We combine the nontrivial relations obtained so far.

```
> SandNrels:=‘union‘(psirels,convert(agb,set),orthrel);
```

$SandNrels := \{v\,(-u\,f + 2\,v\,f + d - 1),\ -2\,d + f^2 - 2\,g^2,\ u\,t - v\,t + u\,w - 3\,v\,w,$
$2 + t^2 + 2\,u\,v - 4\,v^2 - w^2,\ -2\,d\,v^2 + d\,u\,v + v\,f + g\,w + t\,g - g^2,$
$1 + d + d\,u\,v - 2\,d\,v^2 + v\,f + 2\,t\,g - g^2,\ -4\,d\,v\,t + d\,u\,w + d\,u\,t + 3\,f\,t + 3\,f\,w + 6\,v\,g - 3\,f\,g,$
$-u\,t\,f + 4\,v\,t\,f - u\,w\,f + 3\,t\,d - 3\,t + 3\,w\,d - 3\,w,\ -2\,u\,v\,f^2 + 4\,v^2\,f^2 + u\,f\,d - u\,f - d^2 + 2\,d - 1,$
$1 + d + 2\,d\,u\,v - 4\,d\,v^2 + u\,f + 2\,t\,g + 2\,g\,w - f^2,\ 2\,t\,v^2 - t^3 - 2\,w\,v^2 - t^2\,w + t\,w^2 + w^3 - 2\,t - 2\,w,$

$16\,v\,u\,t\,f-24\,t\,v^2\,f-u^2\,t\,f-u^2\,w\,f-12\,d\,v\,t+12\,v\,t+3\,d\,u\,w-3\,u\,w+3\,d\,u\,t-3\,u\,t,$
$d\,u^2\,t-16\,d\,v\,u\,t+24\,d\,t\,v^2+d\,u^2\,w-12\,v\,t\,f+3\,u\,w\,f+3\,u\,t\,f+9\,g+18\,g\,u\,v-36\,g\,v^2-9\,d\,g,$
$-6\,u\,f-6\,f\,u^2\,v+4\,f\,u\,t^2+4\,f\,t\,w\,u+12\,f\,v^2\,u-16\,f\,t^2\,v+18\,d-18+18\,d\,u\,v-18\,u\,v-36\,d\,v^2$
$+36\,v^2,\,6\,d\,u+6\,d\,u^2\,v-4\,d\,u\,t^2-4\,d\,t\,w\,u-12\,d\,v^2\,u+16\,d\,t^2\,v+9\,f+18\,u\,v\,f-36\,v^2\,f$
$-24\,g\,v\,t+6\,g\,u\,w+6\,g\,u\,t-9\,d\,f,\,27+6\,d\,u^2+6\,d\,u^3\,v-8\,d\,u^2\,t^2-8\,d\,w\,u^2\,t-12\,d\,u^2\,v^2$
$+80\,d\,u\,t^2\,v-96\,d\,t^2\,v^2+18\,u\,f+18\,f\,u^2\,v-12\,f\,u\,t^2-12\,f\,t\,w\,u-36\,f\,v^2\,u+48\,f\,t^2\,v+6\,g\,u^2\,t$
$-96\,g\,v\,u\,t+144\,g\,t\,v^2+6\,g\,u^2\,w-27\,d^2,\,-6\,u^2\,f-6\,u^3\,v\,f+8\,u^2\,t^2\,f+8\,w\,u^2\,t\,f+12\,u^2\,v^2\,f$
$-80\,u\,t^2\,v\,f+96\,t^2\,v^2\,f+18\,d\,u-18\,u+18\,d\,u^2\,v-18\,u^2\,v-12\,d\,u\,t^2+12\,u\,t^2-12\,d\,t\,w\,u$
$+12\,t\,w\,u-36\,d\,v^2\,u+36\,v^2\,u+48\,d\,t^2\,v-48\,t^2\,v\}$

```
> indets(SandNrels);
```

$$\{d, f, g, t, u, v, w\}$$

```
> SandNbasis:=factor(Basis(SandNrels,plex(d, f, g, u, v,t, w)));
```

$SandNbasis := [2\,t\,v^2-t^3-2\,w\,v^2-t^2\,w+t\,w^2+w^3-2\,t-2\,w,\,u\,t-v\,t+u\,w-3\,v\,w,$
$2+t^2+2\,u\,v-4\,v^2-w^2,$
$-2+t^2-4\,v^2+3\,w^2+2\,g^2-2\,g\,w-g\,w^3-6\,t\,g-g\,t^3+w\,g\,t^2+g\,w^2\,t+4\,t\,w,$
$2\,f-4\,v-w^2\,f+t^2\,f-2\,g\,v\,t+2\,g\,w\,v,\,t^2-2\,v^2+w^2-g\,w+v\,f-t\,g+2\,t\,w,$
$2+2\,t^2-4\,v^2+2\,w^2+u\,f-3\,g\,w-t\,g+4\,t\,w,$
$-v\,t+f\,g-g\,w^2\,v-f\,t-f\,w-v\,g\,t^2+2\,g\,v\,t\,w+v\,w-2\,v\,g,$
$t^2+4\,t\,w+3\,w^2-4\,v^2-4\,t\,g-4\,g\,w-g\,t^3+w\,g\,t^2+g\,w^2\,t-g\,w^3+f^2,\,-g\,w+1+t\,g+d]$

```
>
```

The next step is to introduce the relations involving the roots of unity theta_i. The first goal is to show that th2 satisfies a nontrivial degree 3 (or less) polynomial in Q[d].

```
> psith0:=Vector([1,d*th1,f*th2,g*th3,g*th3]):
> Thetarels:=simplify(subs({},[th1^2*S[2,2]-(sum('M1[k,2]*psith0[k]'
  ,k=1..5)),th1*th2*S[2,3]-(sum('M1[k,3]*psith0[k]',k=1..5)),th1*th3
  *S[2,4]-(sum('M1[k,4]*psith0[k]',k=1..5)),th3*th2*S[4,3]-(sum('M2[
  k,4]*psith0[k]',k=1..5)),th2^2*S[3,3]-(sum('M2[k,3]*psith0[k]',k=1
  ..5)),th3^2*(S[4,4]+S[4,5])-(sum('(M3[k,4]+M3[k,5])*psith0[k]',k=1
  ..5))]));
```

$Thetarels := [th1^2-1-n1\,d\,th1-n2\,f\,th2-2\,n3\,g\,th3,\,-th1\,th2\,f-n2\,d\,th1-n4\,f\,th2-2\,n5\,g\,th3,$
$th1\,th3\,g-n3\,d\,th1-n5\,f\,th2-n6\,g\,th3-n7\,g\,th3,\,-n5\,d\,th1-n9\,f\,th2-n10\,g\,th3-n11\,g\,th3,$
$th2^2\,d-th2^2-1-n4\,d\,th1-n8\,f\,th2-2\,n9\,g\,th3,\,-th3^2\,d-th3^2-1-d\,th1\,n6-d\,th1\,n7$
$-f\,th2\,n10-f\,th2\,n11-2\,g\,th3\,n12-g\,th3\,n13-g\,th3\,n14]$

```
>
```

we immediately find that th1=1.

```
> factor(Basis(`union`(SandNrels,convert(subs(rules1,Thetarels),set)
  ),tdeg(th1,u, v,t, w,d,th2,th3,f,g)))[9];
```

$$g\,th3\,(th1-1)$$

```
> Thetarels1:=factor(numer(subs({th1=1},Thetarels)));
```

$Thetarels1 := [-n1\, d - n2\, f\, th2 - 2\, n3\, g\, th3, -f\, th2 - n2\, d - n4\, f\, th2 - 2\, n5\, g\, th3,$
$g\, th3 - n3\, d - n5\, f\, th2 - n6\, g\, th3 - n7\, g\, th3, -n5\, d - n9\, f\, th2 - n10\, g\, th3 - n11\, g\, th3,$
$th2^2\, d - th2^2 - 1 - n4\, d - n8\, f\, th2 - 2\, n9\, g\, th3,$
$-th3^2\, d - th3^2 - 1 - d\, n6 - d\, n7 - f\, th2\, n10 - f\, th2\, n11 - 2\, g\, th3\, n12 - g\, th3\, n13 - g\, th3\, n14]$

```
> indets(Thetarels1);
```

$$\{d, f, g, n1, n10, n11, n12, n13, n14, n2, n3, n4, n5, n6, n7, n8, n9, th2, th3\}$$

```
> map(degree,Thetarels1,th3);
```

$$[1, 1, 1, 1, 1, 2]$$

```
> solve(Thetarels1[2],th3);
```

$$-\frac{f\, th2 + n2\, d + n4\, f\, th2}{2\, n5\, g}$$

We substitute back in to eliminate th3 and take at the numerators of the resulting rational functions, which give us our new relations.

```
> Thetarels2:=factor(numer(subs({th3=solve(Thetarels1[2],th3)},Theta
  rels1)));
```

$Thetarels2 := [-n1\, d\, n5 - n5\, f\, th2\, n2 + n3\, f\, th2 + n3\, n2\, d + n3\, n4\, f\, th2, 0, -f\, th2 - n2\, d - n4\, f\, th2$
$- 2\, n3\, d\, n5 - 2\, n5^2\, f\, th2 + n6\, f\, th2 + n6\, n2\, d + n6\, n4\, f\, th2 + n7\, f\, th2 + n7\, n2\, d + n7\, n4\, f\, th2,$
$-2\, n5^2\, d - 2\, n9\, f\, th2\, n5 + f\, th2\, n10 + n10\, n2\, d + n10\, n4\, f\, th2 + f\, th2\, n11 + n11\, n2\, d + n11\, n4\, f\, th2,$
$th2^2\, d\, n5 - th2^2\, n5 - n5 - n4\, d\, n5 - n8\, f\, th2\, n5 + n9\, f\, th2 + n9\, n2\, d + n9\, n4\, f\, th2, -n4^2\, f^2\, th2^2$
$- 2\, n4\, f^2\, th2^2 - 2\, d\, n4\, f^2\, th2^2 - 2\, d^2\, n4\, f\, th2\, n2 - 2\, d^2\, f\, th2\, n2 + 4\, n12\, n5\, g^2\, n4\, f\, th2$
$+ 2\, n13\, n5\, g^2\, n4\, f\, th2 + 2\, n14\, n5\, g^2\, n4\, f\, th2 - 2\, f\, th2\, n2\, d - n2^2\, d^3 - n2^2\, d^2 - 4\, n5^2\, g^2$
$- 4\, d\, n7\, n5^2\, g^2 - 4\, d\, n6\, n5^2\, g^2 - d\, n4^2\, f^2\, th2^2 + 4\, n12\, n5\, g^2\, f\, th2 + 4\, n12\, n5\, g^2\, n2\, d$
$+ 2\, n13\, n5\, g^2\, f\, th2 + 2\, n13\, n5\, g^2\, n2\, d + 2\, n14\, n5\, g^2\, f\, th2 + 2\, n14\, n5\, g^2\, n2\, d - th2^2\, f^2$
$- 2\, n4\, f\, th2\, n2\, d - d\, f^2\, th2^2 - 4\, f\, th2\, n10\, n5^2\, g^2 - 4\, f\, th2\, n11\, n5^2\, g^2]$

```
> Thetarels2red:=factor(Reduce(subs(rules1,Thetarels2),SandNbasis,pl
  ex(d, f, g, u, v,t, w,th2))):

> map(degree,Thetarels2red,th2);
```

$$[-\infty, -\infty, -\infty, -\infty, 2, 2]$$

```
> factor([coeff(Thetarels2red[5],th2,2),coeff(Thetarels2red[5],th2,1
  ),coeff(Thetarels2red[5],th2,0)]);
```

$[9\, v\, (-w + t)\, (t\, g - g\, w + 2), -9\, g\, w^2\, v - 18\, v\, t + 18\, f\, w + 18\, f\, t - 9\, v\, g\, t^2 + 18\, g\, v\, t\, w + 18\, v\, w,$
$9\, v\, (-w + t)\, (t\, g - g\, w + 2)]$

provided v(t-w) is not zero, th2 satisfies a nonzero degree 2 polynomial (observe that if g is rational, so is d which is a contradiction). The following shows that if v(t-w)=0, then d=1.

```
> factor(Basis(‘union‘(SandNrels,{v*(t-w)}),plex(f,g,u,v,t,w,d)));
```

$[(d - 1)\, (d + 1), w\, (d - 1), t\, (d - 1), -w^2 + 1 + t^2 + d, v\, (d + 1), v\, w, v\, t, u\, (d + 1), u\, w, u\, t,$
$2\, u\, v + 1 - 4\, v^2 - d, d\, g + g - 2\, w - 2\, t, (-w + g - t)\, w, t\, g + 1 - t\, w - w^2 + d,$
$2\, g^2 + 1 - 4\, v^2 - 4\, t\, w - 4\, w^2 + 3\, d, d\, f - f + 4\, v, v\, (-2\, v + f), u\, f + 1 - 4\, v^2 - d, f\, g - 2\, v\, g - f\, w - f\, t,$

$f^2+1-4v^2-4tw-4w^2+d]$

```
> with(numtheory):
```

Since [Q[d,f,g]:Q]=2, th2 satisfies a cyclotomic polynomial of degree 1,2 or 4.

```
> invphi(4); invphi(2);
```

$$[5, 8, 10, 12]$$

$$[3, 4, 6]$$

```
> ThetaElim:=subs(rules1,‘union‘(convert(Thetarels,set),{th1-1}));
```

$$ThetaElim := \{th1-1, -\left(-\frac{4vt}{3}+\frac{uw}{3}+\frac{ut}{3}\right)d\,th1-(t+w)f\,th2-2vg\,th3, -th1\,th2\,f - \left(\frac{2u}{3}+\frac{2u^2v}{3}-\frac{4ut^2}{9}-\frac{4twu}{9}-\frac{4v^2u}{3}+\frac{16t^2v}{9}\right)d\,th1-(1+2uv-4v^2)f\,th2 - 2\left(-\frac{4vt}{3}+\frac{uw}{3}+\frac{ut}{3}\right)g\,th3, th1^2-1 - \left(\frac{2u^2}{9}+\frac{2u^3v}{9}-\frac{8u^2t^2}{27}-\frac{8wu^2t}{27}-\frac{4u^2v^2}{9}+\frac{80ut^2v}{27}-\frac{32t^2v^2}{9}\right)d\,th1 - \left(\frac{2u}{3}+\frac{2u^2v}{3}-\frac{4ut^2}{9}-\frac{4twu}{9}-\frac{4v^2u}{3}+\frac{16t^2v}{9}\right)f\,th2 - 2\left(\frac{u^2t}{9}-\frac{16vut}{9}+\frac{8tv^2}{3}+\frac{u^2w}{9}\right)g\,th3, th1\,th3\,g-\left(\frac{u^2t}{9}-\frac{16vut}{9}+\frac{8tv^2}{3}+\frac{u^2w}{9}\right)d\,th1 - \left(-\frac{4vt}{3}+\frac{uw}{3}+\frac{ut}{3}\right)f\,th2-(1+uv-2v^2)g\,th3-(-2v^2+uv)g\,th3, th2^2d-th2^2-1-(1+2uv-4v^2)d\,th1-uf\,th2-2(t+w)g\,th3, -th3^2d-th3^2-1-d\,th1(-2v^2+uv)-d\,th1(1+uv-2v^2)-2f\,th2\,v-3g\,th3\,t-g\,th3\,w\}$$

```
> indets(ThetaElim);
```

$$\{d, f, g, t, th1, th2, th3, u, v, w\}$$

```
> bases1:=i->factor(Basis(‘union‘(SandNrels,ThetaElim,{cyclotomic(i,
  th2)}),plex(d,f,g,th1,th3,th2,u,v,t,w)))[1];
```

bases1 := *i* → factor(*Groebner*:-*Basis*(‘union‘(*SandNrels*, *ThetaElim*, {*numtheory*:-*cyclotomic*(*i*, *th2*)}), plex(*d*, *f*, *g*, *th1*, *th3*, *th2*, *u*, *v*, *t*, *w*)))$_1$

```
> map(bases1,invphi(4));
```

$$[(22201-19008w^6+20736w^8-201456w^4-65532w^2)(t^2-2tw+4+w^2), (-36481-89856w^6+10368w^8+159984w^4+19704w^2)(t^2-2tw+4+w^2), (-58081-302400w^6+103680w^8+187920w^4+75660w^2)(t^2-2tw+4+w^2), (6w^2+18w+13)(6w^2-18w+13)(288w^4+24w^2-121)(t^2-2tw+4+w^2)]$$

```
> map(bases1,invphi(2));
```

$$[(9w^2+4)(t^2-2tw+4+w^2), (-1+60w^2+72w^4)(t^2-2tw+4+w^2), t^2-2tw+4+w^2]$$

clearly have no solutions.

```
> expand((t-w)^2+4);
```

$$t^2 - 2\,t\,w + 4 + w^2$$

```
> factor(Basis(`union`(SandNrels,ThetaElim,{th2^2-1}),plex(f,g,th1,th3,th2,u,v,t,w,d)));
```

$[(d-1)^2\,(d+1)^2,\ (d-1)\,(d+1)\,(25\,d+25+128\,w^2),\ (d-1)\,(d+1)\,(14\,w\,d+19\,w+15\,t),$

$-(d+1)\,(d^2-2\,d-3+2\,t\,w-t^2-w^2),$

$75\,t+52\,w-15\,t\,d+32\,w\,d+15\,w^3-7\,d^2\,w-17\,d^3\,w+15\,t^3-15\,t^2\,w-15\,t\,w^2,$

$2-2\,t^2+8\,v^2-6\,w^2-3\,d+d^3-8\,t\,w,$

$29\,v-6\,u+6\,d\,u-13\,v\,d-12\,w\,t\,v+6\,t^2\,v+7\,d^2\,v+6\,v\,w^2+v\,d^3,\ -219\,v+64\,u+64\,u\,w^2$

$-16\,w^2\,v\,d+53\,v\,d+80\,w\,t\,v-32\,t^2\,v+64\,v\,w^3\,t-32\,t^2\,w^2\,v+16\,d\,w\,t\,v-32\,v\,w^4-5\,d^2\,v$

$-304\,v\,w^2-21\,v\,d^3,\ u\,t-v\,t+u\,w-3\,v\,w,\ 6+4\,u\,v-3\,d+d^3-8\,w^2-8\,t\,w,$

$(d-1)\,(d^2+2\,d+2\,th2-1),\ (t^2-2\,t\,w+4+w^2)\,(th2-1),\ (th2-1)\,(th2+1),$

$(d+1)\,(3\,d^2-2\,d-5+4\,th3),\ 28\,v-6\,u+4\,v\,d-3\,th3\,v\,t^2+6\,th3\,v\,t\,w-6\,w\,t\,v+3\,t^2\,v+6\,th3\,u$

$-24\,th3\,v-3\,th3\,v\,w^2-4\,d^2\,v+3\,v\,w^2-4\,v\,d^3,\ 3+d-d^3-d^2-2\,th2-2\,th3+2\,th2\,th3,\ th1-1,$

$(d+1)\,(7\,d^2\,w+10\,w\,d+30\,g-15\,t-2\,w),$

$-3+3\,d+4\,w^2-3\,d^3-4\,t\,w+3\,d^2+16\,g\,w+4\,w^2\,d-4\,d\,w\,t,$

$13+19\,d+4\,w^2-3\,d^3-4\,t\,w+3\,d^2+16\,t\,g+4\,w^2\,d-4\,d\,w\,t,\ 9\,v\,t-6\,w^3\,v-52\,v\,w-48\,v\,g$

$-6\,v\,w\,t^2+12\,t\,w^2\,v+12\,g\,u-5\,w\,d^2\,v-6\,v\,w\,d-3\,v\,d^3\,w+9\,d\,v\,t+12\,u\,w,$

$-10\,g+5\,t-w-4\,d^2\,w+4\,w\,d-4\,d^3\,w+10\,th2\,g-5\,th2\,t+5\,th2\,w,$

$13\,w+13\,w\,d-5\,t-8\,d^2\,w-5\,t\,d-8\,d^3\,w+20\,g\,th3,\ 2\,d+1+d^2+4\,g^2,$

$-11\,v-5\,v\,d-2\,v\,w^2+3\,d^2\,v+4\,w\,t\,v-2\,t^2\,v+5\,v\,d^3+4\,f]$

```
>
```